*Black, Owens, Wedeking & Wohlfarth
*Chapter 5 Replication


use "Ch-5_Data_Emotions.dta"

*Table 5.1

*Model 1
logit jvotePet c.affectP##c.consciPI c.affectR##c.consciPI openPI extraPI agreePI neuroPI, robust

*Model 2
logit jvotePet c.affectP##c.consciPI c.affectR##c.consciPI openPI extraPI agreePI neuroPI lcDisagreement petAuthority respAuthority usBriefP usBriefR  usAmPet usAmResp petNumStat respNumStat attyOAexpP_log attyOAexpR_log numberQuestions_petnr numberQuestions_respnd totPetAm totRespAm  colemanLiau_readabilityP colemanLiau_readabilityR lowLib scIdeo lowLibXscIdeo, robust

*Figure 5.1
quietly margins, at(affectP=(1(1)9) consciPI=(-1.9 0.90) (median))
marginsplot, x(affectP) xlabel(1(1)9) plotopt(msymbol(none)) noci xtitle(Percent Emotional Brief Content) ytitle(Probability Justice Votes for Petitioner) title("Petitioner's Brief") scheme(s1mono) legend(order(1 "Low Conscientiousness")) saving(pb)
quietly margins, at(affectR=(1(1)10) consciPI=(-1.9 0.90) (median))
marginsplot, x(affectR) xlabel(1(1)9) plotopt(msymbol(none)) noci xtitle(Percent Emotional Brief Content) ytitle(Probability Justice Votes for Petitioner) title("Respondent's Brief") scheme(s1mono) legend(order(2 "High Conscientiousness")) saving(rb)
graph combine pb.gph rb.gph, scheme(s1mono) ycommon

*Figure 5.2
quietly margins, dydx(affectP affectR) at(consciPI=(-1.9(0.1)0.9) (median))
marginsplot, xlabel(-1.9(.4).9) ylabel(-.06(.03).06) recastci(rarea) plotopt(msymbol(none)) ciopts(color(gs10)) ylin(0) xtitle(Justice Conscientiousness) ytitle("Marginal Effect of Emotive Content" "on Petitioner Vote Probability") title("") scheme(s1mono)


clear


use "Ch-5_Data_OA.dta"

*Table 5.2

*Model 1
logit vote ideodiffmq complexity c.grade_zmean##c.consciPI openPI extraPI agreePI neuroPI if justice ~="blkm", robust cl(justice)

*Model 2
logit vote c.consciPI##c.grade_zmean c.ideodiffmq##c.grade_zmean c.complexity##c.grade_zmean openPI extraPI agreePI neuroPI us_appellant us_appellee sg_argue_applt sg_argue_applee dcprac_notus_applt dcprac_notus_appellee lawprof_applt lawprof_applee clerk_applt clerk_applee elite_law_applt elite_law_applee prev_exp_dif_log  if justice ~="blkm", robust cl(justice)

*Figure 5.3
quietly margins, at(grade_zmean=(-4.5(0.5)4.5) consciPI=(-3.67 0.90) (median))
marginsplot, xdim(grade_zmean) xlabel(-4.5(1)4.5) plotopt(msymbol(none)) noci  xtitle(Petitioner Oral Argument Performance Advantage) ytitle(Probabilty Justice Votes for Petitioner) title("") scheme(s1mono) legend(order(1 "Low Conscientiousness" 2 "High Conscientiousness"))

*Figure 5.4
quietly margins, dydx(grade_zmean) at(consciPI=(-3.68(0.5)0.90) (median)) level(90)
marginsplot, xdim(consciPI) xtitle("Justice Conscientiousness") ytitle("Marginal Effect of Petitioner Performance"  "Increase on Petitioner Vote Probability") title("") scheme(s1mono) plotopt(msymbol(none)) recastci(rarea) ciopts(color(gs10)) ylin(0)

